
Введение

Естественные вычисления • Тематика на-
стоящего издания связана с проблемами построе-
ния и визуализации так называемых естественных
вычислительных моделей. Естественные вычисле-
ния1 — это относительно новый раздел приклад- 1 Англ. — Natural Computing.

ной науки, образовавшийся на стыке математики,
информатики и естественных наук (прежде всего
биологии). Под естественными вычислениями при-
нято понимать различные математические и ком-
пьютерные модели, прообразом которых являют-
ся разнообразные природные системы и процессы.
Сюда входят как классические модели, например
клеточные автоматы, нейронные сети и генетиче-
ские алгоритмы, так и относительно новые, в част-
ности модели роевого интеллекта.

Отличительной особенностью практически всех
такого рода моделей является их распределенность,
или многоагентность. Каждая модель состоит, как
правило, из большого числа относительно просто
устроенных объектов — агентов. Сложное нетриви-

РИС. 1 Системный эффект
альное поведение моделей обеспечивается взаимо-
действием этих простых объектов, это называется
системным эффектом (поведение системы сложнее
поведения ее частей, рис. 1), или эмерджентно-
стью. Исследование эмерджентных свойств систем
до сих пор остается фундаментальной проблемой
в разных областях науки.

Среда моделирования NetLogo • Методы
аналитического исследования естественных вычис-
лительных моделей, как правило, являются очень
скудными. Основным методом их изучения оказы-
вается компьютерное моделирование. Отметим, что
во многих случаях интересующий исследователей

модели эффект достигается только при очень боль-
шом числе входящих в нее объектов. Это, в свою
очередь, определяет серьезные требования к про-
граммному и аппаратному обеспечению процесса
моделирования. Однако для первоначального озна-
комления с данной тематикой вполне достаточно
иметь какую-нибудь простую систему прототипи-
рования многоагентных моделей.

РИС. 2 Ури Виленски

В качестве такой системы в настоящем учебни-
ке предлагается использовать среду многоагентно-
го моделирования NetLogo, разработанную в кон-
це 1990-х годов американским ученым и педаго-
гом Ури Виленски2. В основу этой свободно рас-2 U. Wilensky, W. Rand, An

introduction to agent-based
modeling: Modeling natural, social
and engineered complex systems
with NetLogo, Cambridge: MIT
Press, 2015.

пространяемой среды положен язык программиро-
вания NetLogo — отдаленный потомок языка че-
репашьей графики Logo, разработанного с обра-
зовательными целями еще в 1960-х годах одним
из классиков искусственного интеллекта Сеймуром
Пейпертом3.3 По состоянию на июнь 2010 года

насчитывалось порядка 250 реа-
лизаций Logo с момента создания
языка!

РИС. 3 Сеймур Пейперт

Вселенная в модели NetLogo состоит из двумер-
ного поля, поделенного на отдельные клетки — пат-
чи, по которым перемещаются разного рода аген-
ты, называемые черепахами (рис. 4). Последние мо-
гут взаимодействовать как между собой, так и с
окружающей их средой.

РИС. 4 Модель NetLogo «Wolf Sheep Predation»

Неотъемлемой чертой процесса построения мо-
делей в NetLogo является их визуализация, разра-
ботчик модели может управлять в динамике боль-
шим количеством визуальных атрибутов как чере-
пах, так и патчей. Кроме того, среда NetLogo под-
держивает построение разного рода графиков, что
делает ее полноценной средой численного исследо-
вания поведения разрабатываемых моделей.

Структура книги • Каждая глава издания по-
священа рассмотрению какой-нибудь одной моде-
ли (либо семейства моделей). В первых парагра-
фах главы дается теоретическое описание модели,
после чего идет подробное, фактически пошаговое
описание процесса построения этой модели в среде
NetLogo4. 4 Для разработки моделей ав-

тором использовалась версия 6.0
среды NetLogo.В конце каждой главы приводится список за-

даний для самостоятельной проработки изложен-
ного в главе материала: технические упражнения
по доработке построенной модели, задания по чис-
ленному исследованию поведения данной модели
и «творческие» задания по разработке и визуали-
зации аналогичных моделей.

Изложение материала учебника сопровождает-
ся большим числом иллюстраций, таблиц, графи-
ков и библиографических ссылок. В приложении
к учебнику дается краткое описание всех конструк-
ций языка NetLogo, использовавшихся при построе-
нии моделей. Более полная документация доступна
через систему помощи среды или на официальном
сайте NetLogo. Крайне полезным также будет изу-
чение моделей из большой библиотеки готовых мо-
делей NetLogo, включающей в себя, в частности,
примеры использования многих конструкций язы-
ка и интерфейса NetLogo.

Автор с благодарностью примет конструктив-
ные или доброжелательные комментарии, отзывы
и пожелания, которые читатель сможет оставить
по адресу netlogo.tutorial@gmail.com.





ГЛАВА 1
Задача Бюффона

Методы Монте-Карло • Семейство методов
Монте-Карло представляет собой собрание разно-
образных методов и моделей, объединенных иде-
ей проведения большой серии стохастических вы-
числительных экспериментов с последующей ста-
тистической обработкой их результатов. Причем

РИС. 1.1 Станислав Улам

чаще всего под статистической обработкой понима-
ется вычисление математического ожидания и, воз-
можно, среднеквадратичного отклонения. Автором
метода Монте-Карло (как вычислительного алго-
ритма) является американский математик Стани-
слав Улам.

Идея алгоритма пришла Уламу в то время, ког-
да он лежал в больнице и занимался раскладыва-
нием карточного пасьянса Кэнфилд (аналог нашей
Косынки). Он попытался найти вероятность того,

РИС. 1.2 Начальная конфигу-
рация в пасьянсе Кэнфилд

что этот пасьянс сойдется. Однако эта задача ока-
залась слишком сложной для методов теории ве-
роятностей. Тогда ему и пришла в голову мысль
применить следующий подход — надо разложить
пасьянс n раз и определить количество m тех слу-
чаев, когда пасьянс сошелся. Тогда отношениеm/n
(так называемое выборочное среднее) и даст при-
ближенную оценку искомой вероятности. Этот ал-
горитм был практически сразу применен к задаче
вычисления многомерных определенных интегра-
лов; заметим, что эта задача до сих пор остается
одним из важнейших применений метода Монте-

РИС. 1.3 Николас Метрополис

Карло, мы рассмотрим ее подробнее в следующей
главе.

Название метода было придумано Николасом
Метрополисом, коллегой Улама, как утверждает-

ся, в честь его дяди — страстного игрока в рулетку.
В 1949 году они с Уламом опубликовали совмест-
ную статью, которая так и называлась — «Метод
Монте-Карло»1.1 N. Metropolis, S. Ulam, The

Monte Carlo Method, Journal of the
American Statistical Association,
Vol. 44, No. 247, 1949, p. 335–341.

Задачи, решаемые методами Монте-Карло, мо-
гут быть как вероятностными, так и детермини-
рованными. В первом случае метод Монте-Карло
применяется, например, для того, чтобы численно
определить те или иные характеристики распреде-
ления интересующей нас случайной величины, ког-
да аналитически это сделать или очень сложно, или
даже невозможно. Именно такую задачу и рассмат-
ривал Улам, когда придумал данный метод.

РИС. 1.4 Жорж-Луи Леклерк,
граф де Бюффон

Тем не менее методы Монте-Карло применяют-
ся и для решения детерминированных задач. Идея
состоит в том, что нужная нам (детерминирован-
ная) величина связывается с некоторыми статисти-
ческими параметрами, найденными при проведе-
нии серии случайных экспериментов. К этому клас-
су задач как раз и относится вычисление опреде-
ленных интегралов (т.е. площадей, объемов и т.д.).
Классической же иллюстрацией такого подхода яв-
ляется задача Бюффона2, которая была поставлена2 Buffon’s needle problem.

и решена задолго до официального появления ме-
тода Монте-Карло.

Задача Бюффона • В этой задаче у нас име-
ется плоская поверхность, на которой проведены

h l

РИС. 1.5 Задача Бюффона

параллельные прямые, отстоящие друг от друга на
заданное расстояние h (рис. 1.5). Простым приме-
ром такой модели может служить дощатый пол или
разлинованный лист бумаги.

На эту поверхность бросаются иголки (мы, од-
нако, будем бросать спички), длина которых l мень-
ше расстояния h. Каждая спичка может пересечь
какую-нибудь прямую (зеленая спичка) или не пе-
ресечь ни одну из них (красная спичка). Вероят-
ность p того, что спичка пересечет некоторую пря-
мую, может быть вычислена аналитически3:3 В нахождении этой вероятности,

собственно, и заключалась ориги-
нальная задача Бюффона.

p =
2l

πh
. (1.1)

Если теперь провести большую серию таких экс-
периментов по бросанию спичек и подсчитать час-
тоту (выборочное среднее) p̂ = n/m тех случаев,

когда спички пересекают какую-нибудь линию, то
с помощью формулы 1.1 оказывается возможным
найти приближенное значение числа π4: 4 Хотя сам Буффон такой опыт не

проводил (см. E. Behrends, Buffon
— Did he actually throw sticks?
Newsletter of the EMS, 2014, June,
p. 47–49), опыт с бросанием иго-
лок в XIX веке неоднократно про-
водился. Например, в 1864 го-
ду некий капитан О.С. Фокс, вы-
полнив около 500 бросаний иг-
лы, нашел приближенное значе-
ние π ≈ 3.1423 (см. A. Hall, On
an experimental determination of
Pi, The Messenger of Mathematics,
Vol. 2, 1872, p. 113–114).

π ≈ π̂ =
2l

hp̂
=

2ln

hm
. (1.2)

Особенно удобно пользоваться этой формулой, ког-
да длина спички ровно в два раза меньше расстоя-
ния h. В этом случае число π приблизительно оце-
нивается как n/m.

В таблице 1.1 приведены результаты простого
компьютерного эксперимента по решению задачи
Бюффона методом Монте-Карло. Эта таблица хо-
рошо демонстрирует тот факт, что методы Монте-
Карло, при всей их универсальности, обладают су-
щественным недостатком — скорость их сходимо-
сти к точному решению, как правило, очень низ-
кая. В частности, погрешность между выборочным

ТАБЛ. 1.1 Сходимость метода
Монте-Карло в задаче Бюф-
фона

n π̂

10 4.00000
100 3.01886
1000 3.00751
10 000 3.15768
100 000 3.14545
1000 000 3.14233
10000 000 3.14041
100000 000 3.14157

средним и математическим ожиданием с ростом n
ведет себя как 1/

√
n (рис. 1.6). Это означает, что

для получения решений с высокой точностью тре-
буется проведение очень большого числа экспери-
ментов.

n

E
1
0
−

5
1
0
−

4
1
0
−

3

2×107 4×107 6×107 8×107

РИС. 1.6 Сравнение экспери-
ментальной (голубая линия)
и аналитической (красная
линия) погрешности мето-
да Монте-Карло в задаче
Бюффона

При этом отметим, что методы Монте-Карло
легко и эффективно распараллеливаются, что от-
части решает проблему их медленной сходимости.
Например, в задаче Бюффона все бросания можно
выполнять одновременно, т.к. они являются взаим-
но независимыми. Именно этот прием мы и приме-
ним при построении нашей первой распределенной
модели в среде NetLogo.

Построение модели • Наша цель — смоде-
лировать задачу Бюффона, выполняя броски за-
данного числа спичек на разлинованную горизон-
тальными линиями поверхность. Модель должна
показывать результат каждого броска, раскраши-
вая головку спички в красный или зеленый цвет
(рис. 1.5). По результатам бросков спичек должна
выполняться оценка числа π. Кроме того, интер-
фейс модели должен содержать графики сходимо-
сти метода (рис. 1.6).

A Создаем новую модель (меню File → New).

B Через диалог модели (по клику правой кнопки
мыши на черном поле) настраиваем ее параметры:

начало координат помещаем в левый нижний угол
(Location of origin → Corner, Bottom Left), размеры
max-pxcor и max-pycor полагаем равными 3 и 5 соот-
ветственно, размер патча Patch size — 100 пикселям
(рис. 1.7).

РИС. 1.7 Диалог с настройкой
параметров модели

C Добавляем к интерфейсу кнопку setup, для это-
го нажимаем на кнопку Add на панеле инструмен-
тов при выбранном типе элемента Button (справа
от кнопки Add) и указываем место расположения
новой кнопки. Сразу же открывается диалог с на-
стройкой параметров новой кнопки (рис. 1.8). В по-
ле Commands указываем команду setup, которая
будет вызываться при нажатии этой кнопки. Эта
же строка будет по умолчанию и текстом кнопки5.

5 Можно явным образом указать
текст на кнопке, если ввести его
в поле Display name в диалоге на-
стройки.

В результате в нашем интерфейсе появится новая

РИС. 1.8 Настройка парамет-
ров кнопки

кнопка, но цвет ее текста будет красным — это зна-
чит, что связанная с ней команда пока нами не
определена.

D Переходим во вкладку Code и пишем код про-
цедуры setup для настройки начальной конфигу-
рации модели.

1 to setup
2 clear-all
3 ask patches [
4 ifelse pycor mod 2 = 0
5 [set pcolor blue]
6 [set pcolor sky]
7 ]
8 reset-ticks
9 end

В первой строке мы объявляем новую процедуру
setup с помощью ключевого слова to. Первой ко-
мандой процедуры setup практически всегда явля-
ется команда clear-all (строка 2), которая очища-
ет модель от результатов предыдущих запусков6.6 В частности, при этом происхо-

дит обнуление всех переменных. После этого происходит настройка всех патчей с по-
мощью команды ask patches — при выполнении
этой команды происходит перебор в некотором слу-
чайном порядке всех патчей модели и для каждого
патча выполняются действия, указанные в квад-
ратных скобках команды. В нашем случае мы про-
веряем (команда ifelse, строка 5), в какой строке
располагается данный патч (с четной или нечет-
ной y-координатой, pycor mod 2 = 0), и в зависи-
мости от этого просим установить (команда set,

строки 6 и 7) цвет текущего патча (pcolor) либо
синим (blue), либо голубым (sky)7. Последней ко- 7 Цвета в NetLogo представляют-

ся числами от 0 до 140, подобрать
нужный цвет можно с помощью
инструмента Color Swatches (меню
Tools, рис. 1.9).

РИС. 1.9 Цвета в NetLogo

мандой процедуры setup, как правило, идет коман-
да reset-ticks, которая сбрасывает таймер моде-
ли и вызывает ее прорисовку. Код любой процеду-
ры или функции должен заканчиваться ключевым
словом end.

РИС. 1.10 Начальная конфигу-
рация модели

E Переходим во вкладку Interface и проверяем ра-
боту кнопки setup — после нажатия на нее модель
должна принять вид, показанный на рис. 1.10.

F Добавляем к интерфейсу слайдер (slider), ис-
пользуя кнопку Add. В диалоге настройки связы-
ваем слайдер с переменной turtles-number8, ко-

8 Имена переменных в NetLogo
могут содержать в себе практиче-
ски любые символы (кроме ско-
бок).

торая будет обозначать число выбрасываемых за
один раз спичек, в качестве минимального значе-

РИС. 1.11 Настройка слайдера

ния указываем 1 (рис. 1.11). Теперь в любом месте
кода нам будет доступна переменная с указанным
именем.

G Добавляем к интерфейсу кнопку go, связанную
с командой go. При ее настройке включаем пере-
ключатели Disable until ticks start (блокировка кноп-
ки до инициализации таймера, которую мы выпол-
няем в процедуре setup) и Forever (автоматическое
нажатие кнопки после завершения работы связан-
ной с ней команды).

H Создаем процедуру go и пишем ее код, в кото-
ром моделируется бросание заданного количества
спичек.

1 to go
2 create-turtles turtles-number
3 [setup-turtle]
4 tick
5 end

В строке 2 мы с помощью команды create-turtles
создаем заданное количество turtles-number че-
репах, каждую из которых просим выполнить код,
указанный в скобках, в данном случае это проце-
дура setup-turtle, код которой мы определим на
следующем шаге. Команда tick служит для обнов-
ления таймера и вызывает перерисовку модели9. 9 При выбранном режиме показа

on ticks справа в панели инстру-
ментов на вкладке с интерфейсом.
Для всех рассматриваемых нами
моделей рекомендуется использо-
вать именно этот режим.

I Создаем процедуру setup-turtle, в начале ко-
торой определим визуальные характеристики чере-
пахи:

1 set shape "circle"
2 set size 0.1
3 set color black
4 set pen-size 4

В первой строке задаем форму черепахи (shape) —
круг, затем ее размер (size10), цвет (color11) и10 Измеряется в патчах.

11 Это также цвет следа, который
оставляет черепаха при своем пе-
ремещении.

толщину следа (pen-size12).

12 Измеряется в пикселях.

J Саму спичку мы будем рисовать в два приема,
сначала поместим черепаху в случайную точку мо-
дели, потом передвинем в случайном направлении
на заданное расстояние с прорисовкой следа. Та-
ким образом получим отрезок, на одном конце ко-
торого и будет располагаться черепаха (визуально
это будет головка спички, рис. 1.12). Цвет спич-
ки мы должны определить из условия пересечения
черепахой горизонтальной линии между патчами.
Это мы можем сделать, просто сравнив цвета пат-

начальная
точка

конечная
точка

след

РИС. 1.12 Схема рисования од-
ной спички

чей в начальной и конечной точках: если они разли-
чаются, то пересечение есть. Таким образом, вто-
рая часть процедуры setup-turtle будет выгля-
деть следующим образом.

1 setxy random-xcor random-ycor
2 pendown
3 let c1 pcolor
4 fd 0.4
5 let c2 pcolor
6 ifelse c1 = c2
7 [set color red]
8 [set color lime]

При создании черепахи она автоматически поме-
щается в начало координат и ориентируется в слу-
чайном направлении, переместить ее в заданную
точку можно с помощью команды setxy (строка 1),
два параметра которой задают координаты точки.
Случайные координаты точки13 можно получить с13 С учетом реальных размеров

модели. помощью команд random-xcor и random-ycor. Пос-
ле этого командой pendown включаем режим про-
рисовки следа14. Цвет патча, в котором располага-14 Этот режим можно отключить

командой penup. По умолчанию ре-
жим прорисовки следа отключен. ется в данный момент черепаха, определяется ко-

мандой pcolor. В строке 2 мы запоминаем в пе-
ременной c1 цвет в начальной точке, команда let
создает новую локальную переменную. В третьей
строке командой fd15 мы передвигаем черепаху на15 Сокращение от forward, можно

использовать и такой вариант ко-
манды. расстояние 0.4 патча в ее текущем (случайном) на-

правлении. Запоминаем цвет нового патча в пере-
менной c2, после чего сравниваем два цвета (стро-

ка 5). Если цвет патчей одинаковый, то окраши-
ваем черепаху в красный цвет (red), иначе — в
светло-зеленый (lime).

K Переходим в интерфейс модели и проверяем ее
работу, нажимая сначала на кнопку setup, а за-
тем на кнопку go. Чтобы остановить работу мо-
дели, надо еще раз нажать на кнопку go. После
выполнения нескольких итераций модель должна
выглядеть, как на рис. 1.13. Заметим, что по умол-

РИС. 1.13 Вид модели после
нескольких итераций

чанию в NetLogo применяется так называемое цик-
лическое замыкание границ, когда правая граница
фактически оказывается склеенной с левой, а верх-
няя — с нижней. Поэтому в нашей модели некото-
рые черепахи при движении за одну из границ поля
появляются с его другой стороны16. 16 Этот режим можно поменять

в настройках модели с помощью
двух переключателей World wraps
horizontally и World wraps vertically.

L Теперь займемся приближенной оценкой чис-
ла π согласно формуле (1.2). Для этого нам потре-
буются два счетчика — число всех черепах (пере-
менная n) и число черепах, которые при своем дви-
жении пересекли горизонтальную границу между
патчами (переменная m). Создадим соответствую-
щие глобальные переменные в начале нашего кода
с помощью следующей команды.

1 globals [n m]

Аргументом этой команды является список (в квад-
ратных скобках) имен глобальных переменных.

M Инициализацию этих переменных выполним в
процедуре setup с помощью следующих двух ко-
манд.

1 set n 0
2 set m 0

Их надо вставить в любом месте процедуры между
командами clear-all и reset-ticks17, например 17 В принципе, эта инициали-

зация не является обязательной
в нашем случае, т.к. команда
clear-all и так обнуляет все гло-
бальные переменные.

после команды (D:2).

N Наконец, в процедуре setup-turtle, перед ко-
мандой (J:6) с проверкой на равенство двух цветов,
увеличиваем n на 1.

1 set n n + 1

А перед командой (J:8) смены цвета черепахи на
светло-зеленый (внутри квадратных скобок) уве-
личиваем на 1 аналогичным образом и перемен-
ную m.

1 set m m + 1

O Определим три вспомогательные функции (в
терминах NetLogo функции называются репорте-
рами). Функция approx-pi будет вычислять при-
ближенное значение числа π согласно (1.2).

1 to-report approx-pi
2 report 2 * 0.4 * n / m
3 end

Функции, в отличие от процедур, в NetLogo опре-
деляются командой to-report. Значение, возвра-
щаемое функцией, определяется командой report.
В нашем случае (строка 2) возвращаемое значение
задается формулой (1.2) с учетом того, что h = 1 —
расстояние между полосами у нас равно одному
патчу, а l = 0.4 — такой шаг черепахи мы уста-
новили в (J:4).

P Вторая функция model-error вычисляет экс-
периментальную погрешность, т.е. модуль разности
между точным значением числа π и приближенной
оценкой.

1 to-report model-error
2 report abs (approx-pi - pi)
3 end

Модуль числа в NetLogo находится функцией abs.

Q Наконец, функция theory-error будет вычис-
лять теоретическую оценку погрешности 1/

√
n.

1 to-report theory-error
2 report 1 / sqrt n
3 end

Квадратный корень из числа в NetLogo определя-
ется с помощью функции sqrt.

РИС. 1.14 Настройка монитора

R Перейдем на вкладку с интерфейсом и добавим
к нему монитор (monitor), используя кнопку Add.
В открывшемся окне настройки указываем следу-
ющие параметры (рис. 1.14). В поле reporter ука-
зываем выражение approx-pi. В поле Display Name
указываем строку pi18. Проверяем работу модели18 Если оставить это поле пус-

тым, то заголовок монитора бу-
дет содержать выражение из поля
reporter.

(кнопки setup и go) — в новом мониторе будет отоб-
ражаться приближенное значение числа π, которое
постепенно должно становиться все более близким
к точному значению19.19 3.1415926535897932384 — заме-

тим, однако, что такая точность
в нашей модели является недости-
жимой.

S Последним пунктом построения нашей модели
является включение в нее графика сходимости, по-
казывающего, как изменяется погрешность в зави-
симости от числа n черепах (т.е. брошенных спи-

чек). Выбираем тип элемента Plot, нажимаем кноп-
ку Add и выполняем настройку графика (рис. 1.15).
В поле Name (заголовок графика) указываем Error
plot, в поле X axis label (подпись оси горизонтальной
оси) — n, в поле Y axis label (подпись оси вертикаль-
ной оси) — error, в поле Y max (начальное макси-
мальное значение по вертикальной оси) — 1. Вклю-
чаем переключатели Auto scale? (автоматическое
масштабирование графика) и Show legend? (пока-
зывает названия кривых). Далее настраиваем сами

РИС. 1.15 Настройка графика

графики (группа элементов управления Plot pens),
у нас их два — экспериментальная погрешность и
теоретическая погрешность. Выбираем цвет первой
кривой (например, красный). В поле Pen name ука-
зываем имя model, в поле Pen update commands за-
писывается команда рисования:

plotxy n log model-error 10.

Команда plotxy определяет график зависимости
одной величины от другой. Первый ее аргумент (n)
соответствует величине, откладываемой по оси абс-
цисс, второй (log model-error 10, десятичный ло-
гарифм ошибки) — по оси ординат. Если нажать на

РИС. 1.16 Дополнительные на-
стройки кривой

кнопку с изображением карандаша, то можно по-
лучить доступ к дополнительным настройкам кри-
вой (рис. 1.16). Например, можно определить тип
кривой mode как Point. Аналогичным образом на-

РИС. 1.17 Окончательный ин-
терфейс модели

страиваем вторую кривую (предварительно нажав
на кнопку Add pen). Цвет — голубой, имя кривой —
theory, команда рисования:

plotxy n log theory-error 10.

T Проверяем работу модели. Окончательный ин-
терфейс модели должен иметь вид, как на рис. 1.17.

УПРАЖНЕНИЯ

1 Выполните вручную опыт Бюффона. Вместо иго-
лок можно использовать любые линейные объекты —
спички, зубочистки, булавки. В Интернете можно най-
ти описания даже таких экзотических опытов, как бро-
сание замороженных сосисок для хот-догов.

2 Если после 10 итераций погрешность метода Монте-
Карло в задаче Бюффона составила 10−1, то сколь-

ко итераций надо выполнить, чтобы достичь точности
а) 10−2; б) 10−4; в) 10−8?

3 Проведите серию численных экспериментов с раз-
личными размерами модели — числом патчей по верти-
кали и горизонтали20. Влияет ли изменение размеров20 Поля max-pycor и max-pycor в на-

стройках модели. на скорость сходимости метода?

4 Убедитесь экспериментально, что любое замыкание
границ модели существенным образом влияет на при-
ближенную оценку числа π. Причиной этого явления
является особенность поведения черепах в NetLogo, ко-
торые оказались перед одной из границ и следующим
шагом должны пересечь эту границу. Если границы за-
мкнуты, то черепаха по правилам NetLogo остается на
месте, и по нашему методу определения факта пере-
сечения горизонтальной линии она эту линию гаран-
тированно не пересечет. Поэтому число m оказывается
заниженным, а соответствующая оценка числа π — за-
вышенной. Замыкание каких границ сильнее ухудшает
оценку?

5 Наш вариант вычисления приближенного числа π
в (O:2) может приводить к ошибке деления на нуль. Это
будет происходить на первых итерациях, при условии
что первые броски спички оказываются неудачными
(т.е. не пересекающими горизонтальных линий). Тогда
значение m будет оставаться нулевым, а в коде процеду-
ры approx-pi мы выполняем деление на это значение.
Т.к. этот код вызывается только из графика, то ошиб-
ка в его выполнении не будет приводить к остановке
всей модели. Но в любом случае, это ошибочная си-
туация, и ее следует исправить. Простейшим способом
сделать это (с учетом специфики нашей задачи) явля-
ется инициализация значений n и m в процедуре setup
единицами, а не нулями, что будет выглядеть так, как
будто первый бросок спички был успешным, что никак
не должно повлиять на оценку числа π при больших
значениях n.

6 Длина спички в коде нашей модели фиксирована
и равна 0.4. Такого рода константы в программирова-
нии называются магическими21. Одним из способов ре-21 Если наш код будет смотреть

кто-то незнакомый с описанием
всей задачи (или мы сами спу-
стя некоторое время), то он, ско-
рее всего, не сможет уловить связь
между двумя использованиями
этого значения в (J:4) и (O:2), осо-
бенно если в последнем случае мы
заменим 2 * 0.4 на 0.8.

шения этой проблемы является полное комментирова-
ние кода (комментарии в NetLogo начинаются с сим-
вола ; (точка с запятой) и продолжаются до конца
строки). Однако лучшим решением будет вообще избе-
гать такого рода констант в коде, а использовать име-
нованные константы или переменные. В нашем слу-
чае нужно ввести дополнительную глобальную пере-
менную match-length, например с помощью команды

globals, настроить ее в процедуре setup и затем ис-
пользовать там, где встречается данная константа. Бо-
лее предпочтительный вариант — добавить к интер-
фейсу еще один слайдер с именем match-length, на-
строить его пределы и значение по умолчанию, что ав-
томатически даст нам глобальную переменную с таким
именем и сделает ненужным ее инициализацию в про-
цедуре setup.

7 Формула (1.1) корректно описывает вероятность пе-
ресечения одной из параллельных линий только при
условии l < h (проверьте экспериментально). Если же

+0

+1
+2

РИС. 1.18 Учет числа пересече-
ний для случая длинной иглы

длина иглы больше расстояния между прямыми, то эта
формула будет верной, только если при подсчете пере-
сечений учитывать полное их количество (двойное пе-
ресечение — плюс два к счетчикуm и т.д., см. рис. 1.18).
Для реализации такой модели надо работать не с цве-
том патча, на котором располагается черепаха, а с ее
y-координатой (ycor). Дополнительно надо учесть, что
при движении за верхнюю границу черепаха оказыва-
ется снизу, и наоборот. Для этого будет полезным ат-
рибут dy, указывающий, в частности, на направление
движения черепахи (положительное значение — дви-
жение вверх, отрицательное — вниз).

8 В обобщении Лапласа задачи Бюффона22 поверх- 22 B. J. Arnow, On Laplace’s
Extension of the Buffon Needle
Problem, The College Mathematics
Journal, Vol. 25, No. 1, Jan. 1994,
p. 40–43.

ность разлинована горизонтальными и вертикальными
линиями, расстояния между которыми равны a и b со-
ответственно. В предположении, что игла короче обеих

РИС. 1.19 Модель Бюффона-
Лапласа

длин a и b, вероятность пересечения ею при случайном
броске хотя бы одной линии выражается формулой

p =
2l(a+ b)− l2

πab
. (1.3)

Адаптируйте построенную нами модель задачи Бюф-
фона под задачу Бюффона-Лапласа. Для этого, напри-
мер, надо раскрасить патчи в шахматном порядке с ис-
пользованием четырех цветов (почему?) и учесть, что
в этом случае a = b = 1 (рис. 1.19).

9 Еще одним интересным обобщением задачи Бюф-
фона об игле (needle) является задача Бюффона о лап-
ше (noodle)23. Оказывается, что формула (1.1) остает- 23 J. F. Ramaley, Buffon’s Noodle

Problem // The American Mathe-
matical Monthly, Vol. 76, No. 8
(Oct., 1969), p. 916–918.

ся корректной даже для случая одной иглы длины l
произвольной формы. Т.е. для приближенного опреде-
ления числа π достаточно бросить на разлинованную
поверхность одну (n = 1) длинную кривую (ту самую
лапшу) и подсчитать количество m ее пересечений всех
заданных линий. Реализуйте описанную вариацию за-
дачи Бюффона. За основу можно взять классический



22 Глава 1

вариант, рассмотренный в главе, и произвести в ней
следующие изменения. К списку глобальных перемен-
ных добавить переменные dl (шаг черепахи, инициа-
лизировать его небольшим значением, например 0.1, в
процедуре setup) и l (полная длина пути, пройденного
черепахами, инициализировать нулем), переменную n
из кода удалить. Создание черепах перенести в setup,
в процедуре setup-turtle оставить только первую ее
часть, отвечающую за определение визуальных харак-
теристик черепахи. В процедуре go сделать вызов про-
цедуры move-turtle, за основу кода которой взять код
второй части «старой» процедуры setup-turtle. При

РИС. 1.20 Внешний вид модели
Buffon’s noodle

этом надо изменить команду движения черепахи впе-
ред на dl, обновить значение l и после этого выпол-
нить случайный поворот вправо на угол от −30◦ до 30◦

с помощью команды right24 random25 60 - 30. Чтобы

24 Вместо right можно использо-
вать короткий вариант rt.

25 Функция random num возвраща-
ет случайное целое число от 0 до
num (не включительно).

корректно отображать цвет при пересечении горизон-
тальной линии (рис. 1.20), определение цвета c2 и сме-
ну цвета черепахи нужно производить до ее движения.
Цвет c2 можно определить с помощью команды

let c2 [pcolor] of patch-ahead dl.

Функция patch-ahead dl возвращает патч, куда долж-
на попасть черепаха, пройдя расстояние dl в ее теку-
щем направлении. Конструкция [a] of X используется
для получения значения атрибута a агента X. Послед-
ним шагом надо удалить кривую theory из графика мо-
дели, а для кривой model изменить команду рисования
на plot log model-error 10.

10 Интересное приложение задачи Бюффона описа-
но в работе26, в которой авторы экспериментально по-26 E. B. Mallon, N. R. Franks,

Ants Estimate Area Using Buffon’s
Needle, Proceedings of the Royal
Society B: Biological Sciences
267.1445, 2000, p. 765–770.

казали, что муравьи вида Leptothorax при поиске мес-
та для нового гнезда оценивают площадь помещений с
помощью приближенного решения модифицированной
задачи Бюффона. Оказывается, что площадь S плос-
кой фигуры можно оценить путем подсчета числа n
пересечений двух нитей длины a и b, расположенных
случайным образом в границах этой фигуры:

S ≈ 2ab

πn
. (1.4)

Мы вернемся к реализации этой модели позже при рас-
смотрении моделей поведения муравьев, основанных на
применении ими феромонов.


